
  کار با ساختار بانک اطلاعاتی
  

  
  مقدمه

  
مجموعه این آیتمها به عنـوان سـاختار   . یک بانک اطلاعاتی حاوی تعدادی جدول، ایندکس، قيود و چيزهای دیگری است          

 راههای ساده ای را برای شما مهيـا کـرده انـد، روش مبنـی بـر      ADO و DAOهر دو روش . پایگاه داده شناخته می شود   
 همان خوبی که اشياء جدیدی در پایگاه داده می سازد، بازیابی اطلاعاتی درباره مشخصات اشيای    شی برای هر دو به    

  . موجود را بازیابی می کند
  ساخت یک بانک اطلاعاتی

کد زیر یک بانک اطلاعاتی . قبل از اینکه جداول یا اشياء دیگر را بتوانيم تعریف کنيم، باید بانک اطلاعاتی را ساخته باشيم
Microsoft Jetجدید را ساخته و باز می کند :  

  
DAO:  

Sub DAOCreateDatabase() 
   Dim db As DAO.Database 
 
   Set db = DBEngine.CreateDatabase(".\New.mdb", dbLangGeneral) 
End Sub 

ADOX:  
Sub ADOCreateDatabase() 
   Dim cat As New ADOX.Catalog 
 
   cat.Create "Provider=Microsoft.Jet.OLEDB.4.0; Data Source=.\New.mdb;" 
End Sub 
 

.  اسـت ADOX در Catalog معـادل شـی   DAO از Databaseهمانطور که ممکـن اسـت خودتـان حـدس زده باشـيد شـی           
 Catalog شـی  Create شـما بایـد از متـد    ADOX با استفاده از Jetبنابراین برای ساختن یک بانک اطلاعاتی جدید از نوع         

  .  کنيداستفاده
 بـصورت صـریح   Locale پـارامتر  ADOXدر کـد   .  مشخص شـده اسـت     dbLangGeneral با   Locale قبلی پارامتر    DAOدر کد   

بـرای  .  اسـت dbLangGeneral معـادل  Microsoft Jet برای فراهم کننده Localeمقدار پيش فرض . مشخص نشده است
  . استفاده کنيدLocale Identifier می توانيد از خاصيت Localeتغيير مقدار 

 اغلــب مــی تــوانيم پــارمتر اختيــاری ســوم را هــم در نظــر بگيــریم کــه مــشخص کننــده  DAO از CreateDatabaseدر متـد  
برای مثال، کد زیر برای ساخت یک بانک اطلاعاتی رمز گـذاری  . اطلاعاتی برای رمزگذاری و نسخه بانک اطلاعاتی است 

  .شده است استفاده Microsoft Jet 1.1شده از نسخه 
  

Set db = DBEngine.CreateDatabase(".\New.mdb", dbLangGeneral,dbEncrypt Or dbVersion11) 
  
.  رمز گذاری و اطلاعات نسخه بانک اطلاعاتی با استفاده از خواص مخصوص فراهم کننده مـشخص مـی شـوند          ADO در

 )رمزگذاری بانـک اطلاعـاتی   (Encrypt Database و )نوع موتور(Engine Type، از خواص Microsoft Jetبرای فراهم کننده 
  :استفاده می شود

   cat.Create "Provider=Microsoft.Jet.OLEDB.4.0;" & "Data Source=.\New.mdb;" & _ 
      "Jet OLEDB:Encrypt Database=True;" & _ 
      "Jet OLEDB:Engine Type=2;" 

  بازیابی اطلاعات الگوی بانک اطلاعاتی
  

 هر دو دارای مجموعـه ای از اشـياء هـستند کـه مـی تواننـد بـرای بازیـابی اطلاعـات الگـوی بانـک                  ADO و هم    DAOهم  
  . مرور مجموعه ها، تعيين کردن ساختار اشيای بانک اطلاعاتی آسان است. اطلاعاتی بکار روند

  
 دارد را بـا اسـتفاده از حلقـه هـا و     کدی که مشاهده می کنيد چگونگی چاپ نام هر جدولی که در بانک اطلاعـاتی قـرار              

  . را تشریح می کندADOX از Tables و مجموعه DAO از TableDefsمجموعه 
  

DAO:   
Sub DAOListTables() 
   Dim db As DAO.Database 
   Dim tbl As DAO.TableDef 
 
   Set db = DBEngine.OpenDatabase(".\NorthWind.mdb") 
   For Each tbl In db.TableDefs 
      Debug.Print tbl.Name 
   Next 
End Sub 

  
ADOX:  

Sub ADOListTables() 
   Dim cat As New ADOX.Catalog 
   Dim tbl As ADOX.Table 
 
   cat.ActiveConnection = "Provider=Microsoft.Jet.OLEDB.4.0;" & _ 
      "Data Source=.\NorthWind.mdb;" 
 
   For Each tbl In cat.Tables 
      If tbl.Type <> "VIEW" Then Debug.Print tbl.Name 
   Next 
End Sub 
 

 برای هر TableDef حاوی یک شی TableDefs ارائه کننده یک جدول در بانک اطلاعاتی است و مجموعه TableDefشی 
 ارائـه کننـده یـک جـدول و     Table ساده تر است، به این صورت که شـی      ADOاین کار با    . جدول در بانک اطلاعاتی است    

  . حاوی همه جداول می باشدTablesمجموعه 
 

 Table ممکن است حـاوی اشـيای   Microsoft Access برای بانک اطلاعاتی ADO در Tables مجموعه DAOالبته برخلاف 
 ADO که در Microsoft Jetبرای مثال بازگشت های خطی، پرس و جوهای غير پارامتری . باشد که جداول واقعی نباشند

 یـک جـدول در   Table اینکـه آیـا شـی    برای شناسایی.  هستندTables ها مطرح شده اند اغلب در مجموعه  Viewبا نام   
 را Typeجدول زیر مقـادیر ممکـن بـرای خاصـيت     .  استفاده می کنيم  Typeبانک اطلاعاتی را ارایه داده است، از خاصيت         

  . استفاده می کنيد ليست کرده استMicrosoft Jet با فراهم کننده ADOوقتی که از 
  

 توضيحات نوع
ACCESS TABLE  شیTableستم  یک جدول سيMicrosoft Accessاست  

LINK  شیTable یک جدول متصل شده از یک منبع داده غير ODBCاست  
PASS-THROUGH  شیTable یک جدول متصل شده از یک منبع داده ODBCاست  
SYSTEM TABLE  شیTable یک جدول سيستم Microsoft Jetاست  

TABLE  شیTableیک جدول است  
VIEW  شیTableشت خطی، پرس و جوی غير پارامتری است یک بازگ 

  
 بازیابی کنيد، می توانيد از متد ADOXعلاوه بر اینکه می توانيد اطلاعات الگوی بانک اطلاعاتی را بوسيله مجموعه های      

OpenSchemaبرای بازگرداندن یک رکوردست حاوی اطلاعاتی درباره جداول بانک اطلاعاتی استفاده کنيد .  
  

 باید متحمل سربار ADOX سریعتر از استفاده از مجموعه ها درون حلقه است، زیرا OpenSchemaز متد عموما استفاده ا
 را OpenSchemaکدی که مشاهده می کنيـد چگـونگی اسـتفادهاز متـد     . ساخت اشيا برای هر عنصر در مجموعه شود       

  :آمد را تشریح می کند بدست ADOX و DAOبرای چاپ همان اطلاعاتی که در مثالهای قبل با استفاده از 
  

Sub ADOListTables2() 
   Dim cnn As New ADODB.Connection 

   Dim rst As ADODB.Recordset 
 
   cnn.Open "Provider=Microsoft.Jet.OLEDB.4.0;" & _ 
      "Data Source=.\NorthWind.mdb;" 
 
   Set rst = cnn.OpenSchema(adSchemaTables) 
 
   Do Until rst.EOF 
      If rst.Fields("TABLE_TYPE") <> "VIEW" Then 
         Debug.Print rst.Fields("TABLE_NAME") 
      End If 
      rst.MoveNext 
   Loop 
End Sub 

  ساختن و تغيير دادن جداول
  

جدول محلی است که در آن تعریف و نوع اول .  می توانند دارای دو نوع از جداول باشندMicrosoft Jetبانکهای اطلاعاتی 
نـوع دوم جـدول متـصل شـده اسـت کـه در آن جـدول در یـک بانـک          . داده هر دو  داخل بانک اطلاعاتی ذخيره می شوند        

  .اطلاعاتی خارجی قرار دارد اما یک اتصال به موازات یک کپی از تعریف جدول در بانک اطلاعاتی ذخيره شده است
  

  ساختن جداول محلی
  . می سازندContactsکه مشاهده می کنيد یک جدول محلی جدید به نام مثالهایی 

  
DAO:  

Sub DAOCreateTable() 
   Dim db As DAO.Database 
   Dim tbl As DAO.TableDef 
 
   Set db = DBEngine.OpenDatabase(".\NorthWind.mdb") 
 
   Set tbl = db.CreateTableDef("Contacts") 
 
   With tbl 
      .Fields.Append .CreateField("ContactName", dbText) 
      .Fields.Append .CreateField("ContactTitle", dbText) 
      .Fields.Append .CreateField("Phone", dbText) 
      .Fields.Append .CreateField("Notes", dbMemo) 
      .Fields("Notes").Required = False 
   End With 
 
   db.TableDefs.Append tbl 
   db.Close 
End Sub 
 

ADOX:  
Sub ADOCreateTable() 
   Dim cat As New ADOX.Catalog 
   Dim tbl As New ADOX.Table 
 
   cat.ActiveConnection = "Provider=Microsoft.Jet.OLEDB.4.0;" & _ 
      "Data Source=.\NorthWind.mdb;" 
 
   With tbl 
      .Name = "Contacts" 
      .Columns.Append "ContactName", adVarWChar 
      .Columns.Append "ContactTitle", adVarWChar 
      .Columns.Append "Phone", adVarWChar 
      .Columns.Append "Notes", adLongVarWChar 
      .Columns("Notes").Attributes = adColNullable 
   End With 
 
   cat.Tables.Append tbl 
   Set cat = Nothing 
End Sub 
 

ابتدا ساختن یک .  استفاده می شود شبيه هم هستندADOX و DAOمراحلی که برای ساخت یک جدول با استفاده از     
، و سرانجام اضافه کـردن جـدول   )Column یا Fieldاضافه کردن اشيای (، اضافه کردن ستونها    )Table یا   TableDef(شی  

  .ی تفاوتهای اندکی وجود دارداگرچه مراحل شبيه به هم هستند ولی از نظر نحو. به مجموعه
   

متـد  .  برای سـاخت سـتون قبـل از اضـافه کـردن آن بـه مجموعـه اسـتفاده کنـيم         Create لازم نيست که از متد       ADOXبا  
Appendمی تواند برای هر دو کار ساختن و اضافه کردن ستون بکار رود .  

جـدولی کـه مـشاهده    .  متفاوت اسـت ADOX و DAOشما حتما توجه خواهيد کرد که نامهای انواع داده برای ستونها در       
 بکار ببری و مقایسه آنها با انواع Microsoft Jet که می توانيد برای بانکهای اطلاعاتی DAOمی کنيد چگونگی انواع داده 

  : را به شما نشان می دهدADOXداده در 
  

 ADOXانواع داده  DAOانواع داده 
dbBinary adBinary 

dbBoolean adBoolean 
dbByte adUnsignedTinyInt 

dbCurrency adCurrency 
dbDate adDate 

dbDecimal adNumeric 
dbDouble adDouble 
dbGUID adGUID 

dbInteger adSmallInt 
dbLong adInteger 

dbLongBinary adLongVarBinary 
dbMemo adLongVarWChar 
dbSingle adSingle 
dbText adVarWChar 

  
  

  تغيير یک جدول موجود
وقتی یک جدولی از قبل ساخته شده است شما ممکن است بخواهيد تغييراتـی را در آن مـثلا بـرای اضـافه و یـا حـذف             

  .و یا تازه کردن اطلاعات جدول متصل شده، انجام دهيد) Validation Rule(کردن ستونها، قانون معتبر سازی 
    

ضافه کردن یک ستون جدید با قابليت افزایش خودکار به یک جدول موجود را کدهایی را که مشاهده می کنيد چگونگی ا   
  :شرح می دهند

   
DAO:  

Sub DAOCreateAutoIncrColumn() 
   Dim db As DAO.Database 
   Dim tbl As DAO.TableDef 
   Dim fld As DAO.Field 
 
   Set db = DBEngine.OpenDatabase(".\NorthWind.mdb") 
 
   Set tbl = db.TableDefs("Contacts") 
 
   Set fld = tbl.CreateField("ContactId", dbLong) 
   fld.Attributes = dbAutoIncrField 
 
   tbl.Fields.Append fld 
   db.Close 
End Sub 

 
ADOX:  

Sub ADOCreateAutoIncrColumn() 
   Dim cat As New ADOX.Catalog 
   Dim col As New ADOX.Column 
 
   cat.ActiveConnection = "Provider=Microsoft.Jet.OLEDB.4.0;" & _ 
      "Data Source=.\NorthWind.mdb;" 
 
   With col 
      .Name = "ContactId" 
      .Type = adInteger 
      Set .ParentCatalog = cat 
      .Properties("AutoIncrement") = True 
   End With 

   cat.Tables("Contacts").Columns.Append col 
   Set cat = Nothing 
End Sub 
 

 در AutoIncrement قبـل از اینکـه خاصـيت    Column از شـی  ParentCatalog توجه کنيـد کـه خاصـيت    ADOXدر این مثال    
يت در مجموعـه  در صـورت دسترسـی بـه هـر خاص ـ      .  ست کنيم مقداردهی شده اسـت      True را به    Propertiesمجموعه  

Properties شی columnباید به یک فراهم کننده مرتبط شده باشد .  

  سخت کليدها و ارتباطهای بين جداول
  

. قبلا ساختار یک جدول تعریف شده است، حالا بهتر اسـت کليـدهایی بـرای جـدول و ارتباطهـایی بـين جـداول بـسازیم                  
Microsoft Jetرتباط را برای بهينه کردن پرس و جو ها استفاده می کند اطلاعات تهيه شده در تعریف کليد و ا.   

  
  ساختن یک کليد اصلی

  
یا (این ستون . یک جدول اغلب یک ستون یا ترکيبی از ستونها دارد که بر اساس آن یک ردیف در جدول منحصربفرد است

ف مـی کنيـد، موتـور بانـک     وقتی شما یک کليـد اصـلی تعری ـ    . کليد اصلی یک جدول ناميده می شود      ) ترکيبی از ستونها  
  . یک ایندکس برای تاکيد یکتا بودن کليد می سازدMicrosoft Jetاطلاعاتی 

  
 که در مثال قبل ساخته شده است استفاده می کنـيم، کـد زیـر چگـونگی تعيـين سـتون       Contactsدر این مثال از جدول   

ColumnIDبه عنوان کليد اصلی را شرح می دهد  :  
  

DAO:  
Sub DAOCreatePrimaryKey() 
   Dim db As DAO.Database 
   Dim tbl As DAO.TableDef 
   Dim idx As DAO.Index 
 
   Set db = DBEngine.OpenDatabase(".\NorthWind.mdb") 
   Set tbl = db.TableDefs("Contacts") 
 
   Set idx = tbl.CreateIndex("PrimaryKey") 
   idx.Primary = True 
   idx.Fields.Append idx.CreateField("ContactId") 
 
   tbl.Indexes.Append idx 
 
   db.Close 
End Sub 
 

ADOX:  
Sub ADOCreatePrimaryKey() 
   Dim cat As New ADOX.Catalog 
   Dim tbl As ADOX.Table 
   Dim pk As New ADOX.Key 
 
   cat.ActiveConnection = "Provider=Microsoft.Jet.OLEDB.4.0;" & _ 
      "Data Source=.\NorthWind.mdb;" 
   Set tbl = cat.Tables("Contacts") 
 
   pk.Name = "PrimaryKey" 
   pk.Type = adKeyPrimary 
   pk.Columns.Append "ContactId" 
 
   tbl.Keys.Append pk 
 
   Set cat = Nothing 
End Sub 
 

کليد اصلی مانند هـر اینـدکس دیگـری سـاخته شـده      . يد اصلی استفاده می شود برای ساختن کل Index شی   DAOبا  
 ودبه خاطر اینکه مبحث کوتاهتر بش: جادوگر. ( تنظيم می شودTrue آن به مقدار Primaryاست با این تفاوت که خاصيت 

د بـه کـار مـی     دارد که برای ساخت کليدهای جدیKey یک شی به نام  ADOاما  )  حذف کردم  اقسمت ساخت ایندکس ر   
 ساخته شد شما باید نوع کليـدی  Keyاما وقتی یک شی . مراحل ساخت یک کليد شبيه به ساخت ایندکس است    . رود

 است که نشان ميدهد شـما مـی خواهيـد یـک     adKeyPrimaryدر اینجا نوع کليد    . را که می خواهيد بسازید تعيين کنيد      
  .کليد اصلی بسازید

  
  :مثال زیر را نگاه کنيد.  برای ساخت و اضافه کردن کليد را در یک خط کد بنویسيدADOXهمچنين می توانيد کد مربوط به 

   pk.Name = "PrimaryKey" 
   pk.Type = adKeyPrimary 
   pk.Columns.Append "ContactId" 

 
   tbl.Keys.Append pk 

  
  :ویسيدبه جای کد بالا می توانيد کد زیر را بن

  
   tbl.Keys.Append "PrimaryKey", adKeyPrimary, "ContactId" 
 

  )کليدهای خارجی(ساخت ارتباطهای یک به چند 
  

جایی که مقـدار کليـد اصـلی در جـدول اصـلی مـی توانـد در چنـدین ردیـف در جـدول             (ارتباطهای یک به چند بين جداول       
یک کليد خـارجی سـتون یـا ترکيبـی از سـتونها      . ر می شوندبوسله ساختن کليدهای خارجی برقرا ) خارجی ظاهر شود  

  .برخلاف کليد اصلی، یک کليد خارجی منحصربفرد نيست. است که مقادیرشان با کليد اصلی جدول دیگر همسان است
  

DAO:  
Sub DAOCreateForeignKey() 
   Dim db As DAO.Database 
   Dim rel As DAO.Relation 
   Dim fld As DAO.Field 
 
   Set db = DBEngine.OpenDatabase(".\NorthWind.mdb") 
 
   db.Relations.Delete "CategoriesProducts" 
 
   Set rel = db.CreateRelation() 
   rel.Name = "CategoriesProducts" 
   rel.Table = "Categories" 
   rel.ForeignTable = "Products" 
 
   Set fld = rel.CreateField("CategoryId") 
 
   fld.ForeignName = "CategoryId" 
 
   rel.Fields.Append fld 
 
   db.Relations.Append rel 
End Sub 
 

ADOX:  
Sub ADOCreateForeignKey() 
   Dim cat As New ADOX.Catalog 
   Dim tbl As ADOX.Table 
   Dim fk As New ADOX.Key 
 
   cat.ActiveConnection = "Provider=Microsoft.Jet.OLEDB.4.0;" & _ 
      "Data Source=.\NorthWind.mdb;" 
 
   Set tbl = cat.Tables("Products") 
 
   tbl.Keys.Delete "CategoriesProducts" 
 
   fk.Name = "CategoriesProducts" 
   fk.Type = adKeyForeign 
   fk.RelatedTable = "Categories" 
 
   fk.Columns.Append "CategoryId" 
 
   fk.Columns("CategoryId").RelatedColumn = "CategoryId" 

   tbl.Keys.Append fk 
 
   Set cat = Nothing 
End Sub 
 

کد .  برای ساختن و اضافه کردن کليد می تواند بصورت خلاصه تر در یک خط کد نوشته شودADOXهمچنين کد مربوط به 
  :ه کنيدزیر را نگا

  
   fk.Name = "CategoriesProducts" 
   fk.Type = adKeyForeign 
   fk.RelatedTable = "Categories" 
 
   fk.Columns.Append "CategoryId" 
 
   fk.Columns("CategoryId").RelatedColumn = "CategoryId" 
 
   tbl.Keys.Append fk 
 

  :معادل آن بصورت خلاصه بدین شکل است
  

   tbl.Keys.Append "CategoriesProducts", adKeyForeign,"CategoryId", "Categories", "CategoryId" 

  ساخت و ویرایش پرس و جوها
  

 و پارامترها و اجـرای پـرس و   SQL است که برای تعيين رشته DAO در QueryDef مشابه شی   ADO از   Commandشی  
  .جو بکار می روند

  
 نمی تواند بصورت مـستقيم بـرای یـک پـرس و جـوی      ADO از Command شی DAO از QueryDefهر چند برخلاف شی  

 QueryDef شاخته می شود با استفاده از تعيين یک نام برای آن، شـی    QueryDefوقتی که شی    . پایدار استفاده شود  
تفـاوت آن بـا شـی    . رد اضافه می شود و در بانک اطلاعاتی قرار مـی گي ـ    QueryDefs بصورت خودکار به مجموعه      DAOاز  

Command  در ADO در این است که همه اشایای Commandشما بایـد بـصورت   .  بصورت پرس و جوهای موقتی هستند
  . برای پایدار کردن آن در بانک اطلاعاتی اضافه کنيدADOX از Procedures یا Views را به مجموعه های Commandصریح 

  
-row( را اگــر بــصورت پــرس و جوهــای برگرداننــده ردیــف  Microsoft Jetای  پــرس و جوهــMicrosoft Jetفــراهم کننــده 

returning (      و بدون پارامتر باشند بصورتView  این فراهم کننده یک روال را بصورت یـک پـرس و جـوی    . ها تعریف می کند
  .گيردیا یک پرس و جوی برگرداننده ردیف پارامتر دار در نظر می ) یک عمل عمده(غير برگرداننده ردیف 

  
  ساختن یک پرس و جوی ذخيره شده

  
  .کدهای زیر چگونگی ساختن یک پرس و جوی برگرداننده ردیف و بدون پارامتر را نشان می دهند

  
DAO:  

Sub DAOCreateQuery() 
   Dim db As DAO.Database 
   Dim qry As DAO.QueryDef 
 
   Set db = DBEngine.OpenDatabase(".\NorthWind.mdb") 
 
   Set qry = db.CreateQueryDef("AllCategories", "SELECT * FROM Categories") 
   db.Close 
End Sub 

  
ADOX:  

Sub ADOCreateQuery() 
   Dim cat As New ADOX.Catalog 
   Dim cmd As New ADODB.Command 
 
   cat.ActiveConnection = "Provider=Microsoft.Jet.OLEDB.4.0;" & "Data Source=.\NorthWind.mdb;" 
 
   cmd.CommandText = "SELECT * FROM Categories" 
   cat.Views.Append "AllCategories", cmd 
 
   Set cat = Nothing 
End Sub 

  
 بـه مجموعـه   ADO از Command استفاده شده بدون پـارامتر و برگرداننـده ردیـف اسـت، شـی      SQLدر این مثال، عبارت     

Views از ADOXتوجه کنيد که وقتی فراهم کننده .  اضافه شده استMicrosoft Jet استفاده می شود، می توانيد شی 
Command  را صــرفنظر از نــوع پــرس و جــویی کــه شــروع بــه ســاختن کــرده ایــم، بــه یکــی از مجموعــه هــای Views یــا 

Procedures  وعـه  هر چند اگر یک پرس و جو مانند ایـن مثـال را بـه مجم      .  اضافه کنيمProcedures    اضـافه کنيـد و سـپس 
شوند، شما در خواهيـد یافـت کـه پـرس و جـو در مجموعـه       ) refresh( دوباره سازی Views و Proceduresمجموعه های   

Procedure قرار ندارد و حالا در مجموعه Viewsاست .  
  

ده ردیف را به یکی از مجموعه های همچنين شما می توانيد یک پرس و جوی پارامتر دار یا یک پرس و جوی غير بازگردانن
Procedures   یا Views  هرچند که   .  اضافه کنيدADOX      در واقع این نوع از پرس و جو ها را در مجموعه Procedures  ذخيـره 

 اضافه و سپس هر دو مجموعه را دوباره سازی کنيد، شما در خواهيـد یافـت      Viewsاگر شما آن را به مجموعه       . می کند 
  . قرار داردProceduresضافه شده حالا در مجموعه که پرس و جوی ا

  
  ساختن یک پرس و جوی پارامتر دار

  
  .کدهای زیر چگونگی ساخت یک پرس و جوی پارامتر دار و ذخيره آن در بانک اطلاعاتی را به شما نشان می دهد

  
DAO:  

Sub DAOCreateParameterizedQuery() 
   Dim db As DAO.Database 
   Dim qry As DAO.QueryDef 
 
   Set db = DBEngine.OpenDatabase(".\NorthWind.mdb") 
 
   Set qry = db.CreateQueryDef("Employees by Region","PARAMETERS [prmRegion] TEXT(255);" & _ 
      "SELECT * FROM Employees WHERE Region = [prmRegion]") 
 
   db.Close 
End Sub 
 

ADOX:  
Sub ADOCreateParameterizedQuery() 
   Dim cat As New ADOX.Catalog 
   Dim cmd As New ADODB.Command 
 
   cat.ActiveConnection = "Provider=Microsoft.Jet.OLEDB.4.0;" & "Data Source=.\NorthWind.mdb;" 
 
   cmd.CommandText = "PARAMETERS [prmRegion] TEXT(255);" & _ 
      "SELECT * FROM Employees WHERE Region = [prmRegion]" 
 
   cat.Procedures.Append "Employees by Region", cmd 
 
   Set cat = Nothing 
End Sub 
 

توجـه کنيـد کـه اگرچـه شـی      .  اسـت DAO و ADOXکد ساختن یک پرس و جوی پارامتر دار بسيار شبيه بـه اسـتفاده از       
Command  از ADOجازه می دهد با استفاده از متـد   به شما اCreateParameter      پارامترهـا را بـسازید امـا ایـن اطلاعـات 

  . تعریف کنيدSQLشما باید پارامترها را بصورت بخشی از رشته . وقتی یک روال ساخته یا به روز شد ذخيره نشده اند
  

 ساخته Microsoft Jet و فراهم کننده ADOX را وقتی یک پرس و جو با SQL عبارت Microsoft Jetهمچنين توجه کنيد که 
 هميـشه گزینـه هـایی از موتـور بانـک      Microsoft Jetفراهم کننده .  تفسير خواهد کردDAOشده است بطور متفاوتی از 

ایـن  . تنظـيم مـی کنـد   ) ANSI( را به خصوصيات مورد قبول موسسه استانداردهای ملی آمریکـا  Microsoft Jetاطلاعاتی 
برای مثال .  در زمان ساخته و یا اجرا شدن پرس و جوها بشودADO و DAO باعث تفاوتهایی در رفتار مسئله ممکن است

  :  در کد قبلی بصورت زیر نوشته شده باشدSQLاگر عبارت 
  

   "PARAMETERS [prmRegion] TEXT;" & _ 
      "SELECT * FROM Employees WHERE Region = [prmRegion]" 

                              
 اسـتفاده مـی کنيـد پـارامتر بایـد یـک فيلـد متنـی         DAOبعد از لغت کليدی، تعيين می کند که وقتی کـه از   ) 255(عبارت  

)dbText ــا ــه از    ) adVarWChar ی ــی ک ــا وقت ــد ، ام ــد    ADOباش ــصورت فيل ــد ب ــی کني ــتفاده م ــا Memo) dbMemo اس  ی
adLongVarWChar (در نظر گرفته می شود.  

  
 بـه دليـل کلمـات رزرو    ADO استفاده مـی شـوند هنگـام اسـتفاده در     DAO که با استفاده از SQLی از عبارات  بعلاوه بعض 

 بـرای  ANSIبرای اطلاعـات بيـشتر در ایـن زمينـه بـه ليـست کلمـات رزرو شـده          . شده اضافی به خطا منجر خواهند شد      
Microsoft Jetمراجعه کنيد  .  

  
  اصلاح یک پرس و جوی ذخيره شده

  
  .زیر چگونگی اصلاح یک پرس و جوی ذخيره شده را به شما نشان می دهندکدهای 

  
DAO 

Sub DAOModifyQuery() 
   Dim db As DAO.Database 
   Dim qry As DAO.QueryDef 
 
   Set db = DBEngine.OpenDatabase(".\NorthWind.mdb") 
 
   Set qry = db.QueryDefs("Employees by Region") 

   qry.SQL = "PARAMETERS [prmRegion] TEXT(255);" & _ 
      "SELECT * FROM Employees WHERE Region = [prmRegion] " & "ORDER BY City" 
 
   db.Close 
End Sub 
 

ADO:  
Sub ADOModifyQuery() 
   Dim cat As New ADOX.Catalog 
   Dim cmd As ADODB.Command 
 
   cat.ActiveConnection = "Provider=Microsoft.Jet.OLEDB.4.0;" & "Data Source=.\NorthWind.mdb;" 
 
   Set cmd = cat.Procedures("Employees by Region").Command 
 
   cmd.CommandText = "PARAMETERS [prmRegion] TEXT(255);" & _ 
      "SELECT * FROM Employees WHERE Region = [prmRegion] " & "ORDER BY City" 
 
   Set cat.Procedures("Employees by Region").Command = cmd 
 
   Set cat = Nothing 
End Sub 
 

 اصلاح شده برای ذخيره تغييرات تنظـيم  Command به شی    Procedure شی   Command خاصيت   ADOدر کد مربوط به     
ایـن تفـاوت از ایـن حقيقـت     . طلاعاتی پایدار نمـی شـد  اگر این مرحله آخر انجام نمی شد تغييرات در بانک ا          . شده است 

 بصورت پرس و جوهـای مـوقتی تنظـيم شـده انـد در حـالی کـه اشـيای         ADO از Commandناشی می شود که اشيای    
QueryDef از DAO   شـما بایــد وقتــی کـه بــا   .  بـصورت پــرس و جوهـای ذخيــره شــده طراحـی شــده انــدCommand هــا و
Procedure ها و Viewی کنيد مراقب این مسئله باشيد ها کار م  .  

